

N=8_Algebraic_proof

July 20, 2021

MULTIPLE EXTREMAL DISC-PACKINGS IN COMPACT HYPERBOLIC SURFACES

** CASE N = 8: ALGEBRAIC PROOF**

An NEC-group uniformizing a non-orientable surface of genus p=3 that contains at least two extremal 3-packings Summary of the particular data for this file. It all comes from the numerical experiments:

- N=8 in 1)
 - A particular set of side-pairing transformations in 6)
 - The two bananas: side-pairing 5, banana 1, sign +1 and side-pairing 7, banana 2, sign -1 in 8)
 - The point $z \sim 0.722+0.189i$ is the first intersection point in 9) (indexed 0, since sagemath first index is 0)
 - The point w in 11) is the center of the disc.
 - A particular set of products of side-pairings in 14)
-

Recall that $u = e^R$ where $\cosh R = \frac{1}{2 \sin \frac{\pi}{N}}$.

Therefore $u + u^{-1} = \frac{u^2 + 1}{u} = \frac{1}{\sin \frac{\pi}{N}}$, thus $\sin \frac{\pi}{N} u^2 - u + \sin \frac{\pi}{N} = 0$ and we have

$$u = \frac{1 + \sqrt{1 - 4 \sin^2 \frac{\pi}{N}}}{2 \sin \frac{\pi}{N}}.$$

```
[1]: #Approximate value of u
N=8
print (e^(arccosh(1/(2*sin(pi/N))))).n(), (( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/
→(2*sin(pi/N))).n()
```

2.14745938013009 2.14745938013009

1) We compute the minimal polynomial of u , assigned to the variable pol .

[2] :

```
N=8  
pol=(( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/(2*sin(pi/N))).minpoly()  
show(pol)
```

```
x^8 - 4*x^6 - 2*x^4 - 4*x^2 + 1
```

[3] : #Approximate value of the roots of pol : notice that u is the largest real root
show($\text{pol}.\text{roots}(\text{ring}=\text{QQbar})$)

```
[(-2.147459380130091?, 1),  
 (-0.4656665496226620?, 1),  
 (0.4656665496226620?, 1),  
 (2.147459380130091?, 1),  
 (-0.5411961001461970? - 0.8408964152537146?*I, 1),  
 (-0.5411961001461970? + 0.8408964152537146?*I, 1),  
 (0.5411961001461970? - 0.8408964152537146?*I, 1),  
 (0.5411961001461970? + 0.8408964152537146?*I, 1)]
```

2) We denote the field $\mathbb{Q}(u)$ as $K1$.

[4] : K1.<u> = NumberField(pol , embedding=2)
CC(u)

#The instruction CC gives the approximate value of an element of the field K1.
#The given embedding into the reals is chosen so that CC(u) gives the right value

[4] : 2.14745938013009

3) Since $u = e^R$, we see that $\sin \frac{\pi}{N}$ is the field element given by $\frac{u}{u^2 + 1}$. We call it sn.

[5] : print(CC(u/(u^2+1)), sin(pi/N).n()) #A numerical verification
sn=u/(u^2+1)
show(sn) #Displays sn as an element of K1

```
(0.382683432365082, 0.382683432365090)
```

```
-1/8*u^7 + 5/8*u^5 - 3/8*u^3 + 7/8*u
```

4) Accordingly, $\cos^2 \frac{\pi}{N} = 1 - \sin^2 \frac{\pi}{N} = 1 - \frac{u^2}{(u^2+1)^2}$ is the field element given by $\frac{u^4+u^2+1}{(u^2+1)^2}$.

We call $v = \cos \frac{\pi}{N}$. It is an element of $\mathbb{Q}(u)$ if and only if $u^4 + u^2 + 1$ is a square in K_1 . We also denote it as cs as a nickname.

We denote $L = \mathbb{Q}(u, v)$. This field agrees with K_1 when $v \in K_1$, of course.

[6] : `print(CC(sqrt(u^4+u^2+1)/(u^2+1)),(cos(pi/N)).n()) #Numerical verification`

```

if (u^4+u^2+1).is_square():
    v=sqrt(u^4+u^2+1)/(u^2+1) # cos(pi/N)
    print(CC(v))
    cs=v
    show(v) #Displays v as an element of K1
    L=K1
else:
    Ring.<t> = PolynomialRing(K1)
    L.<v> = K1.extension(t^2-(u^4+u^2+1)/((u^2+1)^2))
    cs=v

#The code below this line constructs the right embedding of L into the complex
#numbers
#that matches the numerical values we know that u and v must have.
#We call this embedding L_approx
#It will be very important later, in particular in points 8) and 11)
j=0
test=0
while test!=1:
    L_approx = L.embeddings(CC)[j]
    if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-L_approx(v))<10^-3 and
    abs(CC(u)-L_approx(u))<10^(-3): test+=1
    j+=1
print L_approx(u),L_approx(v) #This line must give the values we knew in advance
#(numerical verification)

```

(0.923879532511253, 0.923879532511287)

0.923879532511253

$-3/8*u^7 + 11/8*u^5 + 11/8*u^3 + 13/8*u$

2.14745938013009 0.923879532511302

5) The matrices of a, b, c are

$$a = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \quad b = \begin{pmatrix} 0 & u^{-1} \\ u & 0 \end{pmatrix}, \quad c = \begin{pmatrix} -\cos \frac{\pi}{N} & \sin \frac{\pi}{N} \\ \sin \frac{\pi}{N} & \cos \frac{\pi}{N} \end{pmatrix}$$

```
[7]: a=matrix(2,[1,0,0,-1])
b=matrix(2,[0,u^(-1),u,0])
c=matrix(2,[-cs,sn,sn,cs])
show(a)
show(b)
show(c)
```

```
[ 1  0]
[ 0 -1]

[          0 -u^7 + 4*u^5 + 2*u^3 + 4*u]
[          u                      0]

[ 3/8*u^7 - 11/8*u^5 - 11/8*u^3 - 13/8*u   -1/8*u^7 + 5/8*u^5 - 3/8*u^3 + 7/8*u]
[ -1/8*u^7 + 5/8*u^5 - 3/8*u^3 + 7/8*u -3/8*u^7 + 11/8*u^5 + 11/8*u^3 + 13/8*u]
```

6) We introduce in the list s the side-pairings for this surface group: our conjecture, driven by numerical experiments, is that the surface group thus generated will contain more than one extremal packing.

We use the notation $G_i = (ca)^i(abcb)(ca)^{-i}$.

```
[8]: def G(i):
    return (c*a)^i*(a*b*c*b)*(c*a)^(-i)

s=['zero'] #for s[j] to denote the j-th side-pairing we add an "empty" item in
#position 0
s.append(G(1)^(-2)*(c*a*b)*((c*a)^(-1))*G(1)^4) #s1
s.append(G(2)^2*G(1)^4) #s2
s.append((c*a)^2*(c*a)*a*(c*a)^(-1)*G(1)^(-2)) #s3
s.append(G(2)*G(1)^(-1)*(c*a)*a*(c*a)^(-1)*G(1)^3) #s4
s.append(G(2)^(-2)*(c*a)^2*a) #s5
s.append((c*a)^(-3)*b*(c*a)) #s6
s.append(G(2)^(-3)*(c*a)^4) #s7
s.append(G(2)^(-4)*(c*a)^2*a*(c*a)^4) #s8
s.append(G(1)*(c*a)^(-1)*(c*a)^2*b*(c*a)^(-2)*G(2)) #s9

show(s[1:]) # How do the matrices look like? Their coefficients lie in L, but
#are they too horrible?
```

```
[1]:
[ 9/8*u^7 - 35/8*u^5 - 25/8*u^3 - 37/8*u  17/8*u^7 - 63/8*u^5 - 53/8*u^3 - 77/
-8*u]  [ -1/2*u^6 + 7/4*u^4 + u^2 + 7/4 -5/4*u^6 + 5*u^4 + 11/4*u^2 + 11/2]
[ -5/8*u^7 + 19/8*u^5 + 9/8*u^3 + 25/8*u -11/8*u^7 + 41/8*u^5 + 35/8*u^3 + 55/
-8*u

```

```

[      -1/2*u^2 - 1/2  1/2*u^6 - 2*u^4 - u^2 - 3/2]  [ -9/8*u^7 + 35/8*u^5
+ 25/8*u^3 + 37/8*u -17/8*u^7 + 63/8*u^5 + 53/8*u^3 + 77/8*u]
[      -1/2*u^2 + 1/2 -1/2*u^6 + 2*u^4 + u^2 + 5/2], [ 5/8*u^7 - 19/
-8*u^5 - 9/8*u^3 - 25/8*u 11/8*u^7 - 41/8*u^5 - 35/8*u^3 - 55/8*u],
[      1/2*u^2 + 1/2 -1/2*u^6 + 2*u^4 + u^2 + 3/2]  [ 3/8*u^7 - 13/8*u^5
- 3/8*u^3 - 11/8*u -1/8*u^7 + 3/8*u^5 + 5/8*u^3 + 9/8*u]  [-7/8*u^7 + 27/
-8*u^5 + 19/8*u^3 + 33/8*u 5/8*u^7 - 17/8*u^5 - 21/8*u^3 - 23/8*u]
[      1/2*u^2 - 1/2  1/2*u^6 - 2*u^4 - u^2 - 5/2], [ 1/8*u^7 - 3/
-8*u^5 - 5/8*u^3 - 9/8*u -5/8*u^7 + 19/8*u^5 + 13/8*u^3 + 29/8*u], [ 7/8*u^7 -
-27/8*u^5 - 15/8*u^3 - 37/8*u -7/8*u^7 + 27/8*u^5 + 19/8*u^3 + 33/8*u],
[ 1/2*u^6 - 7/4*u^4 - 2*u^2 - 11/4           1/4*u^4 - 1/2*u^2 + 1/4]
[-1/2*u^6 + 7/4*u^4 + 2*u^2 + 11/4           1/4*u^4 - 1/2*u^2 + 1/4],
[1/2*u^6 - 2*u^4 - u^2 - 5/2 1/2*u^6 - 2*u^4 - u^2 - 3/2]
[      -1/2*u^2 + 1/2           1/2*u^2 + 1/2]
]

```

7) We compute $\cosh^2 \frac{d_n}{2}$ for the distances d_0, d_1, d_2 associated to a tiling of hyperbolic space with regular N-gons of angle $2\pi/3$. We store the three values in the list ch.

First, as $d_0 = 2R$, then $\cosh^2 \frac{d_0}{2} = \cosh R = \frac{(u^2+1)^2}{4u^2}$.

Second, $\cosh d_1 = \cosh^2(d_0) - \sinh^2(d_0) \cos(4\pi/N) = (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)$.

Now, since $\cosh d = 2 \cosh^2 \frac{d}{2} - 1$, we have

$$\cosh^2 \left(\frac{d_1}{2} \right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)}{2}$$

that is

$$\cosh^2 \left(\frac{d_1}{2} \right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(4\pi/N)}{2}.$$

Accordingly, we have $\cosh d_2 = \cosh^2(d_0) - \sinh^2(d_0) \cos(6\pi/N) = (1 - \cos(6\pi/N)) \cosh^2(d_0) + \cos(6\pi/N)$ and therefore

$$\cosh^2 \left(\frac{d_2}{2} \right) = \frac{1 + \cosh d_2}{2} = \frac{1 + (1 - \cos(6\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(6\pi/N)}{2}.$$

NOTE: Recall that $\cos \frac{k\pi}{N}$ is the value at $x = \cos \frac{\pi}{N}$ of the k -th Chebyshev polynomial.

```
[9]: ch=[(u^2+1)^2/(4*u^2)]
ch.append((1+(1-chebyshev_T(4,cs))*(2*ch[0]-1)^2+chebyshev_T(4,cs))/(2))
ch.append((1+(1-chebyshev_T(6,cs))*(2*ch[0]-1)^2+chebyshev_T(6,cs))/(2))

show(ch) # How do these values look like? They lie in L, but are they too
         →horrible?
```

```
[-1/4*u^6 + u^4 + 3/4*u^2 + 3/2,
 -1/2*u^6 + 2*u^4 + 3/2*u^2 + 3,
 -3/4*u^6 + 3*u^4 + 9/4*u^2 + 9/2]
```

8) We compute the equations of the two relevant bananas, and the coordinates of the intersection. We need to recall which bananas are the promising ones in each case (meaning for which side-pairing and which distance d_j) from the information we obtained numerically.

The two variables $sign$ can be (independently) 1 or -1 : we guess the right choice after computing (via CC) the numerical value of the intersection, that we expect to match what we obtained in our numerical experiments.

```
[10]: def H(k,s): #A function that occurs in the equation of the banana
        if s.det() == -1:
            return 4*ch[k]
        elif s.det() == 1:
            return 4*(ch[k] - 1)

def banana(sign, s, k): #The equation itself: choosing sign=+1 or -1 gives each
                        →of the two arcs
    polring=PolynomialRing(L, 'x,y')
    x, y=polring.0, polring.1
    A,B,C,D=s[0][0], s[0][1], s[1][0], s[1][1]
    return
    →sign*((B-C)*(1+x^2+y^2)+2*x*(A-D)-2*y*(B+C))-(1-x^2-y^2)*sqrt(H(k,s)-(A+D)^2+4*det(s))

##### Defining the equations
var('x,y')
sign1,sign2=1,-1 #Choose the right couple of signs to point to the right
                  →component of each banana!
eq1=banana(sign1, s[5], 1) #Insert here the right number of sidepairing s[j]
                           →for this case!!!!!
eq2=banana(sign2, s[7], 2) #The third number is either 0, 1, or 2 and indicates
                           →which banana we want.
                           #Choose the right ones for this case!!!
show(eq1)
```

```

show(eq2)
#####

##### A bit of algebraic geometry here: J is the variety determined by both
→polynomials. We check first that
##### it has dimension 0: it consists of finitely many points. Finally we compute
→those points.
polring=PolynomialRing(L, 'x,y')
J = polring.ideal(eq1, eq2)
print J.dimension()
J.variety(L)

```

```

(-u^6 + 4*u^4 + 3*u^2 + 6)*x + (u^6 - 4*u^4 - 3*u^2 - 2)*y + (-u^6 + 4*u^4 + u^2 +
-4)

(3*u^7 - 11*u^5 - 9*u^3 - 15*u)*y + (1/2*u^7 - 5/2*u^5 + 3/2*u^3 - 7/2*u)

0

```

[10]: [$\{y: 1/4*u^6 - 3/2*u^4 + 7/4*u^2 - 1/2, x: -1/4*u^6 + 2*u^4 - 15/4*u^2\}$]

9) We apply L_approx to the points we have obtained in order to determine which is the one that corresponds to the value we expected from our numerical experiments. If none of them seems to be correct we need to go back to 8) and change the signs.

[11]: #WARNING: The numerical experiments gave $z \sim 0.722+0.189i$ in this surface

```

for point in J.variety(L):
    print L_approx(point[x]),L_approx(point[y])

```

0.721675328781099 0.188504392343353

10) We call zx and zy the real and imaginary part of the right intersection of bananas. The point $z = zx + izy$ is, still conjecturally, the center of a disc of a second (hidden) extremal packing.

[12]: zx, zy=J.variety(L)[0][x], J.variety(L)[0][y]

#Choose J.variety(L)[S] for the right value of S checking the output of 9)

```

show(zx) #Again having a look. Are they very large expressions in u?
show(zy)

```

-1/4*u^6 + 2*u^4 - 15/4*u^2
1/4*u^6 - 3/2*u^4 + 7/4*u^2 - 1/2

11) Our numerical experiments suggest that the following center of the initial k-packing will have an important role:

$$w = \frac{u^2 - 1}{u^2 + 1} \left(\sin \frac{2\pi j}{N} - i \cos \frac{2\pi j}{N} \right)$$

for certain integer j (we need to get the right one in each case from the experiments).

(there is also an special case possible in which $w = 0$, the center of the unit disc).

We compute $\cos \frac{2\pi j}{N}$ as the value at $x = \cos \frac{\pi}{N}$ of the $2j$ -th Chebyshev polynomial. Then, we compute $\sin \frac{2\pi j}{N} = \sqrt{1 - \cos^2 \frac{2\pi j}{N}}$.

Important.- By the last expression we mean *the positive root*, but we don't know which of the two possible roots will Sage associate to the symbols $\sqrt{}$ and $-\sqrt{}$ applied to a given element of the field L. We need again to make use of L_approx to decide which one is the positive one (the warning is motivated for the fact that it may happen that the positive one is the one sage calls $-\sqrt{}$).

```
[13]: w_is_center='yes' #leave uncommented the right line. w_is_center='yes' si w is
         →the center of the unit disc
#w_is_center='no'

if w_is_center=='yes':
    wx=0
    wy=0
elif w_is_center=='no':
    j=3 #Enter here the right value for j in this case
    wy=-(u^2-1)*chebyshev_T(2*j,cs)/(u^2+1)
    wx=(u^2-1)*(sqrt(1-chebyshev_T(2*j,cs)^2))/(u^2+1)
    if L_approx(wx)<0: wx=-wx
    show(wx) #Looking at the expressions as polynomials in u or u and v
    show(wy)
print L_approx(wx), L_approx(wy) #Numerical values
```

12) We define the function $g(z) = \frac{z-w}{\bar{w}z-1} = \frac{(z-w)(w\bar{z}-1)}{|w|^2|z|^2-2re(w\bar{z})+1} = \frac{x+iy-wx-ify}{(wx-ify)(x+iy)-1}$, but rather as a function $\mathbb{R}^2 \rightarrow \mathbb{R}^2$. This is for technical reasons (L has no chosen embedding into the complex numbers, so we couldn't perform in principle products as $u \cdot i$ that we will later need).

We need to compute the point $p = \frac{g(zx+izy)}{1+\sqrt{1-|zx+izy|^2}}$, but the denominator is never in L because $1 - |zx + izy|^2$ is never a square. So we extend the field L to $K2 = L(w)$, where $w = \sqrt{1 - |zx + izy|^2}$.

Finally $m = g(p)$ may be, according to our numerical experiments, the fixed point of an order two elliptic element inducing an automorphism of the surface.

```
[14]: def g(x,y):
    denom=(x*wx+y*wy-1)^2+(y*wx-x*wy)^2
    num1=(x^2+y^2)*wx-x*(wx^2-wy^2)-2*y*wx*wy-x+wx
    num2=(x^2+y^2)*wy-2*x*wx*wy+y*(wx^2-wy^2)-y+wy
    return [num1/denom, num2/denom]

show((1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2)) #Inspecting how does w^2 look like

if (1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2).is_square():
    print 'Warning: apparently w IS in the field L' #Apparently this never
→happens in our files
    #If it happens the last lines of this cell would produce an error, since
→K2_approx
    #wouldn't be defined
else:
    Ring.<y> = PolynomialRing(L)
    K2.<omega> = L.extension(y^2-(1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2))
    px=g(zx,zy)[0]/(1+omega)
    py=g(zx,zy)[1]/(1+omega)
    mx=g(px,py)[0]
    my=g(px,py)[1]
    show(mx) #These are the coordinates of the fixed point of tau as elements of
→the field L.
    show(my)
    #Again choosing a precise embedding of K2 into the complex numbers
    #This is not really needed, but allows the computation of numerical values
→for mx, my
    j=0
    test=0
    while test!=1:
        K2_approx = K2.embeddings(CC)[j]
        if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-K2_approx(v))<10^-3 and
→abs(CC(u)-K2_approx(u))<10^(-3) and K2_approx(omega)>0: test+=1
        j+=1
        print K2_approx(u),K2_approx(v), K2_approx(omega) #This line must give the
→values we knew in advance (numerical verification)

print K2_approx(mx), K2_approx(my)
```

$$11/2*u^6 - 22*u^4 - 33/2*u^2 + 5$$

$$(-9/34*u^6 + 67/68*u^4 + 19/34*u^2 + 77/68)*\text{omega} + 9/34*u^6 - 67/68*u^4 - 19/
\text{34}*u^2 - 77/68$$

```
(-1/17*u^6 + 13/68*u^4 + 4/17*u^2 + 19/68)*omega + 1/17*u^6 - 13/68*u^4 - 4/17*u^2
→- 19/68
```

```
2.14745938013009 0.923879532511283 0.666071177800056
0.433159962429712 0.113143060665790
```

13) We compute the coordinates of $\bar{\tau}$

[15]:

```
tau00=(-2*mx)/(1-mx^2-my^2)
tau01=(2*my+1+mx^2+my^2)/(1-mx^2-my^2)
tau10=(2*my-1-mx^2-my^2)/(1-mx^2-my^2)
tau11=(2*mx)/(1-mx^2-my^2)
tau=matrix(2,[tau00,tau01,tau10,tau11])
show(tau) # Once more we are curious. The entries lie in K2, but: how do they
→ look like?
```

```
[ (-17/56*u^6 + 59/56*u^4 + 53/56*u^2 + 11/8)*omega  (-17/56*u^6 + 69/56*u^4 + 57/
-56*u^2 + 87/56)*omega]
[(27/56*u^6 - 107/56*u^4 - 75/56*u^2 - 129/56)*omega   (17/56*u^6 - 59/56*u^4 -
-53/56*u^2 - 11/8)*omega]
```

14) We create a list `conj` containing what our numerical experiments suggested that should be c_j , the conjugate by τ of the side-pairing s_j , for each j . We finally check that $\tau s_j \tau^{-1} c_j^{-1}$ is trivial.

[16]:

```
conj=['Zero'] #Create this list with the data from the numerical experiments!!!!
conj.append(s[8]^(-1)*s[4]^(-1)*s[8])
conj.append(s[8]^(-1)*s[3]^(-1))
conj.append(s[3]^(-1))
conj.append(s[8]^(-1)*s[4]^(-1)*s[8])
conj.append(s[5]^(-1))
conj.append(s[7]*s[9])
conj.append(s[7]^(-1))
conj.append(s[8]^(-1))
conj.append(s[5])

for j in [1..len(s)-1]:
    show(tau*s[j]*tau^(-1)*conj[j]^(-1))
```

```
[1 0]
[0 1]

[1 0]
[0 1]
```

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

$[]:$